

Implementation of One-Turn Maps in SSCTRK Using ZLIB

S. K. Kauffmann, Y. T. Yan

Superconducting Super Collider Laboratory†
2550 Beckleymeade Ave, Dallas, TX 75237

and

D. M. Ritson

Stanford Linear Accelerator Center‡
Stanford University, Stanford, CA 94309

Abstract. The particle tracking code SSCTRK is currently being adapted to operational simulation and beam-beam effect studies for the Collider rings of the SSC. During beam-beam effect studies, the lattice content of the bending arcs is normally not varied, making fast truncated Taylor map tracking through the arcs an attractive option. The implementation of SSCTRK as a truncated Taylor map tracking program has been carried out using the differential algebra library ZLIB, which simplified the task to that of straightforward translation of SSCTRK kick and drift arithmetic operations to calls to the corresponding polynomial operation subroutines of ZLIB. The accuracy and speed (relative to normal SSCTRK tracking) of truncated Taylor map tracking at 2mm betatron oscillation amplitude was studied in various orders of the map. The seventh order map was found to be in agreement with the normal SSCTRK to about eight significant figures on the first turn, and to a fraction of 1% on the 100,000th turn, for a typical 5cm magnet aperture lattice, and could be made to track at ten times the speed of the normal SSCTRK kick-drift tracking on a scalar architecture (Sun) workstation. (The map tracking subroutines of ZLIB are optimized for vector and parallel architecture supercomputers, and typically achieve even faster relative performance on these, but operational simulation studies will be more conveniently carried out on dedicated workstations which have the incoming generation of "superscalar" CPUs.)

Introduction

The differential algebra library ZLIB¹ offers analytic multivariable polynomial manipulation and truncated Taylor map tracking which permit accelerator Taylor map construction, tracking and analysis to be carried out with ease and efficiency under the framework of an IMSL-style user library. ZLIB is divided into two sublibraries, ZPLIB and TPALIB. ZPLIB is oriented toward speed and efficiency on vectorizing or parallelizing (multitasking) supercomputers, while the more straightforward subroutines of TPALIB generally run a little faster on scalar architecture computers. The fast, thin-concatenated-element, kick-drift tracking code SSCTRK (written for the purpose of tracking proton beams through the Collider rings of the SSC) is in the process of being reconfigured for operational simulation and beam-beam effect studies in the SSC, to be carried out on very fast, dedicated workstations which have the incoming generation of "superscalar" CPUs. As it is usual to hold the lattice content of the bending arcs fixed during beam-beam effect simulation studies, fast tracking with a truncated Taylor map through those arcs at a maximum betatron oscillation amplitude of about 2mm is felt to be an attractive option. Below we discuss the translation of the kick-drift tracking section of SSCTRK into one-turn truncated Taylor map construction code through use of the ZLIB sublibrary TPALIB, with the commented code itself being given

1. Yiton Yan and Chiung-Ying Yan, "ZLIB: A Numerical Library for Differential Algebra (User's Guide)", SSCL-300, July, 1990.

†Operated by the Universities Research Association, Inc., for the U.S. Department of Energy under Contract No. DE-AC02-89ER40486.

‡Under contract with the U.S. Department of Energy, Contract No. DE-AC03-76SF00515.

*To be presented at the IEEE Particle Accelerator Conference,
San Francisco, CA, May 6-9, 1991*

in Appendix A. Code for the analogous translation through use of the ZPLIB sublibrary of ZLIB is given in Appendix B. The accuracy of the various map truncation orders is evaluated vis-a-vis the corresponding SSCTRK kick-drift results at 2mm betatron oscillation amplitude over 100,000 turns for a typical SSC 5cm magnet aperture lattice, and the speed advantage of the acceptably accurate seventh order map on a scalar CPU (Sun) workstation over the original SSCTRK kick-drift code is presented.

Translation of SSCTRK kick-drift tracking to Taylor maps using ZLIB

Unlike the post-TEAPOT tracking program Ztrack, which contains hundreds of statements and involves square roots and trigonometric functions,² the kick-drift tracking section of SSCTRK, consisting of just a few dozen lines of highly optimized, simple code which involve essentially only additions and multiplications, allows economic implementation of one-turn maps. With the help of the ZLIB User's Guide (footnote 1), these arithmetic operations are straightforwardly translated into calls to ZLIB subroutines which perform the corresponding polynomial operations. For example, the basic (FORTRAN coded) vertical kick in SSCTRK,

$PH = PH - RIGITY * IMPOL,$

translates to one polynomial multiplication subroutine call to the TPALIB sublibrary of ZLIB, followed by one polynomial subtraction subroutine call to that sublibrary,

CALL TPAMUL(FRIG,FIMPOL,FTMPO,5) followed by CALL TPASUB(FPH,FTMPO,FPH,NM) .

Notice that the polynomial FRIG corresponds to the quantity RIGITY, the polynomial FIMPOL corresponds to the quantity IMPOL, and the polynomial FPH corresponds to the vertical betatron oscillation angle PH. In addition, we store the intermediate result corresponding to the "kick" product $RIGITY * IMPOL$ in the temporary polynomial FTMPO, which, in the second call, is subtracted from the polynomial FPH to "kick update" it. The last argument "5" of the call to subroutine TPAMUL above is the number of variables of which all of our polynomials are functions, while the last argument NM of the call to subroutine TPASUB above is the number of linearly independent terms in each of our polynomials (this is determined by their number of variables, i.e., 5, and their common order -- note that TPAMUL above will truncate its polynomial multiplication result FTMPO to that common order NO, which needs to have been previously specified in an initial call to the TPALIB "preparation" subroutine TPAPRP(5,NO,NM)).

Not every quantity occurring in the kick-drift tracking section of SSCTRK is translated into a polynomial. Any quantity which remains fixed turn-by-turn around the Collider is treated as a constant in the polynomial algebra. Thus the basic SSCTRK vertical drift,

$Y = Y + L0 * PH,$

contains the fixed drift length L0, which is treated as a constant rather than a polynomial. Of course, the vertical betatron oscillation displacement Y and angle PH are treated as polynomials FY and FPH (all such polynomials are represented in the FORTRAN code as single-index, double-precision arrays of length NM, while the constants are represented as just double-precision unindexed variables). This basic vertical drift is taken care of by a single TPALIB subroutine call which simply updates the polynomial FY as the appropriate linear combination of FY and FPH,

CALL TPALIN(FY,L0,FPH,FY,NM) .

We could manage to do with polynomials of five variables rather than the general accelerator physics set of six variables in this instance because we did not include ripple or noise effects and needed only to update the final

2. The ZLIB map extraction from Ztrack is described in: Yiton Yan, "Zmap: A Differential Algebra Map Extraction Program Using ZLIB", SSCL-299, 1990.

longitudinal displacement after a full turn through the arcs and RF cavity by an additive term which doesn't depend at all on the initial longitudinal displacement (we didn't actually include the RF cavity in our polynomial mapping of the arcs, as it would have cost accuracy for a negligible saving of tracking time). Since the variables of which our polynomials are functions correspond to the irreducible set of turn-by-turn varying initial values at the beginning of each turn, we can effectively drop the longitudinal displacement as such a variable. The change in the longitudinal displacement is, however, calculated as a polynomial depending on the other variables because its value after one turn through the arcs is needed by the RF cavity section of the code. The momentum displacement, on the other hand, must be counted as a variable even though it doesn't change at all in the arcs -- it undergoes a complicated change on every turn in the RF cavity. The remaining four variables are, of course, the vertical and horizontal betatron oscillation displacements and angles. The TPALIB subroutine TPAPOK1 (polyn, const, var no, NM) initializes polyn as const times the variable associated with var no, which indexes the initial value array that is to be used for map tracking. Thus, after initializing TPALIB with CALL TPAPRP(5,NO,NM), we initialize those polynomials which are to begin the map tracking turn as one of the five pure variables having values equal to those of the appropriately indexed members of the initial value array XINPT, e.g.,

```
CALL TPAPOK1(FX,1.0D+0,1,NM); CALL TPAPOK1(FTH,1.0D+0,2,NM);
```

and also

```
CALL TPAPOK1(FY,1.0D+0,3,NM); CALL TPAPOK1(FPH,1.0D+0,4,NM); etc.
```

The initial value array for the beginning of the turn is FORTRAN dimensioned as XINPT(5), and the above statements will initialize the polynomial FX to a particular unit coefficient first-order monomial which will have the value XINPT(1) at the beginning of the turn, the polynomial FTH to another unit coefficient first-order monomial which will have the value XINPT(2) at the beginning of the turn, the polynomial FY to yet another unit coefficient first-order monomial which will have the value XINPT(3) at the beginning of the turn, the polynomial FPH to yet a fourth a unit coefficient first-order monomial which will have the value XINPT(4) at the beginning of the turn, etc.

From the above examples of code we see that the translation of the tracking section of SSCTRK into a polynomial mapping through the use of the TPALIB sublibrary of ZLIB was relatively straightforward and intuitive. The full code produced is given in Appendix A (in the map construction subroutine OPSTPA, the original kick-drift code statements from SSCTRK subroutines OPSTRK and PTRACK are given in commented lines just before their translation into TPALIB subroutine calls), which it may be instructive to study in conjunction with the reading of the ZLIB User's Guide. Two "unofficial" modified TPALIB routines were incorporated into the code, the map writeout (to a file) routine WTPKMAP and the specialized map tracking routine TPKMTRK5, which sacrifices generality in the number of variables and the number of output polynomials (both frozen at five) for maximum map tracking speed. Analogous code for translation using the ZPLIB sublibrary of ZLIB is given in Appendix B.

Comparison of the ZLIB translated map tracking with SSCTRK itself

The one-turn truncated Taylor map construction code given in Appendix A lent itself to a precise check for the correctness of the map in the special case of a purely linear lattice (the number of multipoles PPOLES=1, the relative momentum offset DELP=0, and the RF voltage PRFV0=0, in the language used in subroutines OPSTPA and PRFACC, given in Appendix A). In this instance the results of the first-order one-turn map agreed precisely (essentially to all the digits of the double-precision accuracy used) with the results of the original SSCTRK kick-drift code when tracked through a single turn. This perfect correlation of the purely linear lattice with the corresponding first-order map for one turn is a useful consistency check for the correctness of mapping translations carried out with the aid of ZLIB.

The remaining evaluation of the TPALIB driven mapping translation found in subroutine OPSTPA in Appendix A was done with a typical SSC lattice for 5cm magnet aperture (e.g., the number of multipoles PPOLES=6)

with a realistic value for the relative momentum offset DELP (5×10^{-4}), and the RF voltage turned on. All tracking was done with the invariant horizontal and vertical amplitudes each initialized at 2mm, and various truncation orders of the map were compared with the original SSCTRK kick-drift code result over 100,000 turns. Typically, the combined horizontal-vertical invariant (which is not sensitive to horizontal-vertical coupling) averaged over 5000 turns was monitored. Under these circumstances the second order map behaved unacceptably, with its 5000-turn-averaged combined horizontal-vertical invariant increasing at an accelerating pace with turn number (this invariant was essentially constant for the original SSCTRK kick-drift code), and the particle leaving the beam tube aperture entirely before it had gone 20,000 turns. The third-order map behaved much more reasonably, though it displayed rather the opposite behavior from the second order map -- the horizontal-vertical invariant tended to become smaller with turn number rather than remain constant as it ought to have (in accord with the original SSCTRK kick-drift code behavior), i.e., the particle spiraled inward (see Figure 1). Figure 1 indicates the fourth-order map to have substantially the same behavior as the third-order map, albeit marginally worse. The fifth-order map was a vast improvement (again see Figure 1) and the sixth-order map again showed marginal deterioration relative to the fifth-order (both are sufficiently good that this is hard to follow on Figure 1 without a magnifying glass). This odd-even order "great improvement - marginal deterioration" was a consistent feature of the lattice we were testing at 2mm betatron oscillation amplitudes, but likely was a peculiarity of that lattice rather than a general property of truncated Taylor maps. The "great improvement" of the seventh order map was sufficient to bring the combined horizontal-vertical invariant within one part in 20,000 of that of the original SSCTRK kick-drift code, which difference cannot be resolved within the thickness of the line in Figure 1. Further, on the 100,000th turn itself the horizontal betatron oscillation was still faithful within a fraction of one percent to that of the original SSCTRK kick-drift code, from which the map had, of course, been translated. (After the first turn, the agreement in horizontal betatron oscillation is to about eight significant figures.)




FIGURE 1

Comparison of the combined horizontal-vertical invariant for the truncated Taylor maps of order three through seven with that of the original SSCTRK kick-drift code for initial 2mm amplitude over 100,000 turns.

This very acceptably accurate (for use with the bending arcs in beam-beam effect simulation studies) seventh-order map executed a factor of ten faster on the scalar architecture Sun workstation than did the original SSCTRK kick-drift code when the special purpose map tracking subroutine TPKMTRK5 was used. Subroutine TPKMTRK5, presented in Appendix A, is related to the subroutine TPAMTRK of TPALIB, but sacrifices generality in the number of inputs and outputs (with both frozen at five) to speed. On supercomputers the corresponding subroutines from ZPLIB (see Appendix B), which are optimized for vectorization and multitasking (parallelization), could be expected to have an even greater relative speed advantage.

Conclusion

This use of ZLIB to effect the translation of the tracking section of the SSCTRK kick-drift code to mapping construction and map tracking code shows that ZLIB is indeed a user-friendly, effective, and efficient way to build and track truncated Taylor maps. The unique and powerful vectorization and parallelization features of ZLIB (in the sublibrary ZPLIB) were not used in this particular application, nor were ZLIB's map analysis features, but the easy-to-use, familiar (from its IMSL-like framework), and efficient features of ZLIB made this mapping application one that was easily and quickly carried out.

Appendix A

```

C=====
      SUBROUTINE OPSTPA(IER)
C SIMPLIFIED TRACKING CONTROL FOR SINGLE LATTICE, MULTIPLE COORD
C NO RANDOM GENERATOR CALLS ARE MADE SO THAT SEED00 IS NOT CHANGED
C=====
      INCLUDE 'header'
      INTEGER*4 J,NLIST,LIST(MAXCAS),IER,V
      INTEGER*4 NO,NM,NMR,KK,I
      LOGICAL*4 ONLYTR
      COMMON /ONLTR/ ONLYTR
      CHARACTER*4 JOBID
      COMMON /JBID/ JOBID
C=====NO IS MAP POLYNOMIAL ORDER; SINCE THERE ARE 5 POYNOMIAL VARIABLES,
C=====NM EQUALS (NO+5)!/(5!*NO!), THE POLYNOMIAL ARRAY LENGTH
      PARAMETER (NO=7,NM=792)
C=====POLYNOMIAL VARIABLES (ARRAYS) USED IN MAPPING -- THEY ARE MAINLY
C=====THE CORRESPONDING SSCTRK VARIABLE NAMES WITH LETTER F PREFIXED;
C=====ALSO THE OUTPUT MAP VARIABLE (ARRAY SSCMP) AND THE INPUT AND OUPUT
C=====ARRAYS FOR THE MAP TRACKING (ARRAYS XINPT & YOUTP)
      REAL*8 FX(NM),FTH(NM),FY(NM),FPH(NM),FZMZI(NM),FK1DLP(NM),
+      FK2DLP(NM),FK3DLP(NM),FRIG(NM),FREPOL(NM),FIMPOL(NM),
+      FRETMP(NM),FIMTMP(NM),FTMPO(NM),XINPT(5),
+      YOUTP(5),SSCMP(NM,5)
C=====THE OUTPUT MAP VARIABLE (ARRAY SSCMP) -- EQUIVALENCED TO THE FIVE
C=====POLYNOMIAL OUTPUT VARIABLES
      EQUIVALENCE (SSCMP(1,1),FX(1)),(SSCMP(1,2),FTH(1)),
+      (SSCMP(1,3),FY(1)),(SSCMP(1,4),FPH(1)),(SSCMP(1,5),FZMZI(1))

      IER=0
      IF (PCASES.EQ.0) THEN
          PRINT *, 'XXX NO PCASES XXXX'
          IER=-1
          RETURN
      ENDIF
      WRITE (2,501) 89,PCASES
501  FORMAT(I2,I3,I3,I4,I8,5G12.5)
      NLIST=PCASES
      DO 10 V=1,MAXCAS
          LIST(V)=V
          STEP(V)=0

```

```

10  CONTINUE
C=====INITIALIZE TPALIB FOR MAPPING;
C=====NO INITIALIZATION NEEDED FOR TRACKING ONLY
      IF (.NOT.ONLYTR) THEN
        CALL TPAPRP(5,NO,NMR)
C=====PROTECTION: CONSISTENCY OF NM WITH NMR
      IF (NM.NE.NMR) THEN
        PRINT*,'NM = ',NM,' ', NMR = ',NMR,' ', XX OPSTPA ERROR XX'
        IER=-1
        RETURN
      ENDIF
    ELSE
C=====IF TRACKING ONLY, READ IN MAP FROM FORTRAN UNIT 1 AND
C=====SKIP THE MAPPING SECTION
      CALL RTPAMAP(SSCMP,1,5,NM,1)
      PRINT*,'NO = ',NO,' ', NM = ',NM
      GO TO 200
    ENDIF
C=====MAPPING SECTION (TPALIB TRANSLATION FROM OPSTRK & PTRACK) FOLLOWS;
C=====POLYNOMIALLY TRANSLATED SSCTRK VARIABLES HAVE LETTER F PREFIXED
C=====INITIALIZE X,TH,Y,PH MAP INPUT VARIABLES
      CALL TPAOK1(FX,1.0D+0,1,NM)
      CALL TPAOK1(FTH,1.0D+0,2,NM)
      CALL TPAOK1(FY,1.0D+0,3,NM)
      CALL TPAOK1(FPH,1.0D+0,4,NM)
C=====INITIALIZE ZMZI (Z MINUS Z INITIAL) MAP OUTPUT VARIABLE TO ZERO
      CALL TPAZRO(FZMZI,NM)
C=====INITIALIZE MAP INPUT VARIABLES RELATING TO DELP:
C=====K1DELP(V)=K1*DELP(V)
      CALL TPAOK1(FK1DLP,K1,5,NM)
C=====K2DELP(V)=K2*DELP(V)
      CALL TPAOK1(FK2DLP,K2,5,NM)
C=====K3DELP(V)=K3*DELP(V)
      CALL TPAOK1(FK3DLP,K3,5,NM)
C=====RIGITY(V)=1.D+0/(1.D+0+DELP(V))
      CALL TPAOK1(FRIG,1.0D+0,5,NM)
      CALL TPACADD(1.0D+0,FRIG,FRIG,NM)
      CALL TPAINV(FRIG,FRIG,5)
C=====INITIALIZATIONS ARE DONE.  CREATE MAP BY TRANSLATING PTRACK.
      DO 100 KK=1,POLELS
C=====REPOL(V)=0.D+0
        CALL TPAZRO(FREPOL,NM)
C=====IMPOL(V)=0.D+0
        CALL TPAZRO(FIMPOL,NM)
        DO 120 I=PPOLES,1,-1
C=====RETMP(V)=REPOL(V)+ABRE(I, KK)

```

```

      CALL TPACADD(ABRE(I,KK),FREPOL,FRETMP,NM)
C=====IMTMP(V)=IMPOL(V)+ABIM(I,KK)
      CALL TPACADD(ABIM(I,KK),FIMPOL,FIMTMP,NM)
C=====REPOL(V)=RETMP(V)*X(V)-IMTMP(V)*Y(V)
      CALL TPAMUL(FIMTMP,FY,FTMPO,5)
      CALL TPAMUL(FRETMP,FX,FREPOL,5)
      CALL TPASUB(FREPOL,FTMPO,FREPOL,NM)
C=====IMPOL(V)=IMTMP(V)*X(V)+RETMP(V)*Y(V)
      CALL TPAMUL(FRETMP,FY,FTMPO,5)
      CALL TPAMUL(FIMTMP,FX,FIMPOL,5)
      CALL TPAADD(FIMPOL,FTMPO,FIMPOL,NM)
120    CONTINUE
C=====TH(V)=TH(V)+RIGITY(V)*REPOL(V)
      CALL TPAMUL(FRIG,FREPOL,FTMPO,5)
      CALL TPAADD(FTH,FTMPO,FTH,NM)
C=====PH(V)=PH(V)-RIGITY(V)*IMPOL(V)
      CALL TPAMUL(FRIG,FIMPOL,FTMPO,5)
      CALL TPASUB(FPH,FTMPO,FPH,NM)
C=====Z(V)=Z(V)-X(V)*K2-K1*TH(V)-K3DELP(V)
      CALL TPALIN(FZMZI,-K2,FX,FZMZI,NM)
      CALL TPALIN(FK3DLP,K1,FTH,FTMPO,NM)
      CALL TPASUB(FZMZI,FTMPO,FZMZI,NM)
C=====X(V)=X(V)+L0*TH(V)+K1DELP(V)
      CALL TPALIN(FX,L0,FTH,FX,NM)
      CALL TPAADD(FX,FK1DLP,FX,NM)
C=====Y(V)=Y(V)+L0*PH(V)
      CALL TPALIN(FY,L0,FPH,FY,NM)
C=====TH(V)=TH(V)+K2DELP(V)
      CALL TPAADD(FTH,FK2DLP,FTH,NM)
      IF (MOD(KK,100).EQ.1) PRINT*,'ELEMENTS THRU ',KK,' MAPPED.'
100    CONTINUE
C=====MAPPING IS COMPLETED -- WRITE OUT MAP TO FORTRAN UNIT 1.
      CALL WTPKMAP(SSCMP,1,5,NM,1)
C      CALL SETXY(NLIST,LIST,0)
C =====TRACKING
200    APERR0=.FALSE.
C =====MAP TRACKING
C      PRINT *, 'IN DELP(1),Z(1)', DELP(1),Z(1)
      DO 20 J=1,PTURNS
        DO 30 V=1,PCASES
          XINPT(1)=X(V)
          XINPT(2)=TH(V)
          XINPT(3)=Y(V)
          XINPT(4)=PH(V)
          XINPT(5)=DELP(V)
C=====CALL MAP TRACKING ROUTINE (ONE TURN)

```



```

      CALL TPKMTRK5(SSCMP,NO,XINPT,YOUTP,NM)
      X(V)=YOUTP(1)
      TH(V)=YOUTP(2)
      Y(V)=YOUTP(3)
      PH(V)=YOUTP(4)
      Z(V)=Z(V)+YOUTP(5)
30      CONTINUE
C IF PARTICLES LEAVE APERTURE THE SUBROUTINE RETURNS
      IER=0
      NLIST=0
      CALL PCHEK(NLIST,LIST,J)
      IF (APERRO) THEN
        APERRO=.FALSE.
        IER=1
        PRINT *, 'OUTSIDE APERTURE'
        RETURN
      ENDIF
C DISPERSION FREE COORDINATES AT RF CAVITY
      DO 50 V=1,PCASES
C DISPERSION FREE COORDINATES AFTER HALF QUAD THEN DISPERSION CORR
C IF TRACK IS REVERSED THEN SIGN OF TWISS MUST CHANGE BEFORE AND
C AFTER PRFACC IN XOUT STATEMENTS
        THOUT(V)=TH(V)+QD/2.*X(V)
        PHOUT(V)=PH(V)-QD/2*Y(V)
        XOUT(V)=X(V)-TWISS(3,1,1)*DELP(V)
        YOUT(V)=Y(V)
      WRITE (2,*) J,YOUT(V),XOUT(V)
50      CONTINUE
C CALL PRFACC ADDS RF CAVITY ACCELERATION AND ADIABATIC DAMPING TERM
      CALL PRFACC
      DO 60 V=1,PCASES
C DISPERSED COORDINATES AFTER DISPERSION AND -1/2 QUAD
        X(V)=XOUT(V)+TWISS(3,1,1)*DELP(V)
        Y(V)=YOUT(V)
        TH(V)=THOUT(V)-QD/2.*X(V)
        PH(V)=PHOUT(V)+QD/2*Y(V)
60      CONTINUE
20      CONTINUE
      PRINT *, 'OPSTPA:  X1,TH1,P1,Z1', X(1),TH(1),DELP(1),Z(1)
      RETURN
      END

C=====
      SUBROUTINE PCHEK(NLIST,LIST,J)

```

C THIS SUBROUTINE CHECKS FOR PARTICLES EXCEEDING THE BEAM TUBE APERTURE.
 C RETURNS APERR0=(.GT.APERTURE LIMIT), NLIST=NUMBER OF PARTICLES
 C-OUTSIDE APERTURE ON THIS TURN AND LIST(V)=CASE NUMBERS
 C=====

```

    INCLUDE 'header'
    INTEGER*4 V,NLIST,LIST(MAXCAS),J
    LOGICAL*4 APERR(MAXCAS)
    REAL*8 EMITV(MAXCAS),SQAP,THDUM(MAXCAS),PHDUM(MAXCAS),
+    APX,APTH,APPH,SMEMT(MAXCAS)
    REAL*4 RMSEMT(MAXCAS)
    COMMON /SUMEMT/ SMEMT
    APX=TWISS(2,1,2)/TWISS(2,1,1)
    APTH=APX*TWISS(2,1,1)**2
    APPH=TWISS(2,2,1)**2
    SQAP=2*APLIM**2
    APERR0=.FALSE.
  
```

C=====CHECK WITH VECTORIZED LOGIC IF PARTICLE IS > APLIM

```

    DO 10 V=1,PCASES
      THDUM(V)=TH(V)+QD/2.D+0*X(V)
      PHDUM(V)=PH(V)-QD/2*Y(V)
      EMITV(V)=APX*X(V)**2+APTH*THDUM(V)**2+Y(V)**2+
    >    APPH*PHDUM(V)**2
  
```

C=====ACCUMULATE EMITV FOR EVERY TURNS, THEN PRINT IT OUT AS AN
 C=====RMS AND FLUSH OUT THE ACCUMULATOR.

```

    IF (J.EQ.1) SMEMT(V)=0.0D+0
    SMEMT(V)=SMEMT(V)+EMITV(V)
    IF (MOD(J,EVERY).EQ.0) THEN
      RMSEMT(V)=SQRT(SMEMT(V)/EVERY)
      SMEMT(V)=0.0D+0
      WRITE (2,*) J,V,RMSEMT(V)
    
```

ENDIF

10 CONTINUE

DO 15 V=1,PCASES

APERR(V)=(EMITV(V).GT.SQAP)

15 CONTINUE

DO 20 V=1,PCASES

APERR0=(APERR0.OR.APERR(V))

20 CONTINUE

C =====LABEL IF PARTICLE(S) OUTSIDE LIMIT

```

    IF (APERR0) THEN
      DO 30 V=1,PCASES
        IF (APERR(V)) THEN
          NLIST=NLIST+1
          LIST(NLIST)=V
        
```

ENDIF

30

CONTINUE

```

ENDIF
RETURN
END

```

```

C=====
SUBROUTINE PRFACC
C ORBE0 IS REFERENCE STARTING ENERGY
C ORBE00(V) ARE REFERENCE BEND (MACHINE) ENERGIES AT TIME T.
C ORBP(V) ARE ACTUAL ENERGIES
C DELP(V) IS RELATIVE DIFFERENCE OF ACTUAL ENERGY TO REFERENCE E.
C=====

```

```

INCLUDE 'header'
REAL*8 PDUM(MAXCAS)
INTEGER*4 V

```

```

DO 10 V=1,PCASES
    ORBP(V)=ORBE00(V)*(1.D+0+DELP(V))

    PDUM(V)=ORBP(V)

    ORBP(V)=ORBP(V)+PRFV0*DSIN(2.D+0*PI*Z(V)/LAMBDA)

    DELP(V)=(ORBP(V)-ORBE00(V))/ORBE00(V)
    THOUT(V)=THOUT(V)*PDUM(V)/ORBP(V)
    PHOUT(V)=PHOUT(V)*PDUM(V)/ORBP(V)

```

```

-10 CONTINUE
RETURN
END

```

```

subroutine tpaprp(nv,no,nm)
implicit double precision(a-h,o-z)
parameter (nvm=5,nom=9,nmm=2002,
+          njpm=nmm*nvm,
+          navgm=nom/nvm,nrm=nom-navgm*nvm,
c +          nkpm=nmm*(navgm+2)**nrm*(navgm+1)**(nvm-nrm),
+          nkpm=92378,
+          nmw=nmm*nvm+6)
common /pmul1/ nk1p(nmm)
common /bmul1/ nk1pb(nmm)
common /pmul2/ kp(nkpm)
common /pmul3/ lp(nkpm)
common /pmul4/ iop(njpm)
common /pdrv1/ jd(njpm)
common /pdrv2/ jp(njpm)

```

Appendix B

```

C=====
      SUBROUTINE OPSMAP( IER)
C SIMPLIFIED TRACKING CONTROL FOR SINGLE LATTICE, MULTIPLE COORD
C NO RANDOM GENERATOR CALLS ARE MADE SO THAT SEED00 IS NOT CHANGED
C=====
      INCLUDE 'header'
      INTEGER*4 J,NLIST,LIST(MAXCAS),IER,V
      INTEGER*4 NO,NM,NMR,NOK,KK,I
      LOGICAL*4 ONLYTR
      COMMON /ONLTR/ ONLYTR
      CHARACTER*4 JOBID
      COMMON /JBID/ JOBID
C=====NO IS MAP POLYNOMIAL ORDER; SINCE THERE ARE 5 POYNOMIAL VARIABLES,
C=====NM EQUALS (NO+5)!/(5!*NO!), THE POLYNOMIAL ARRAY LENGTH
      PARAMETER (NO=7,NM=792)
C=====POLYNOMIAL VARIABLES (ARRAYS) USED IN MAPPING -- THEY ARE MAINLY
C=====THE CORRESPONDING SSCTRK VARIABLE NAMES WITH LETTER F PREFIXED;
C=====ALSO THE OUTPUT MAP VARIABLE (ARRAY SSCMP) AND THE INPUT AND OUPUT
C=====ARRAYS FOR THE MAP TRACKING (ARRAYS XINPT & YOUTP)
      REAL*8 FX(NM),FTH(NM),FY(NM),FPH(NM),FZMZI(NM),FK1DLP(NM),
+      FK2DLP(NM),FK3DLP(NM),FRIG(NM),FREPOL(NM),FIMPOL(NM),
+      FRETMP(NM),FIMTMP(NM),FTMPO(NM),XINPT(5),
+      YOUTP(5),SSCMP(NM,5)
C=====THE OUTPUT MAP VARIABLE (ARRAY SSCMP) -- EQUIVALENCED TO THE FIVE
C=====POLYNOMIAL OUTPUT VARIABLES
      EQUIVALENCE (SSCMP(1,1),FX(1)),(SSCMP(1,2),FTH(1)),
+      (SSCMP(1,3),FY(1)),(SSCMP(1,4),FPH(1)),(SSCMP(1,5),FZMZI(1))

      IER=0
      IF (PCASES.EQ.0) THEN
          PRINT *, 'XXX NO PCASES XXXX'
          IER=-1
          RETURN
      ENDIF
      WRITE (2,501) 89,PCASES
501  FORMAT(I2,I3,I3,I4,I8,5G12.5)
      NLIST=PCASES
      DO 10 V=1,MAXCAS
          LIST(V)=V
          STEP(V)=0

```

```

10  CONTINUE
C=====INITIALIZE ZPLIB FOR EITHER TRACKING ONLY OR ELSE MAPPING AS WELL
      IF (ONLYTR) THEN
          CALL ZPTRKP(5,NO,1,NMR)
      ELSE
          CALL ZPPREP(5,NO,1,NMR)
      ENDIF
C=====PROTECTION: CONSISTENCY OF NM WITH NMR
      IF (NM.NE.NMR) THEN
          PRINT*, 'NM = ',NM,' ', NMR = ',NMR,' ', XXX OPSMAP ERROR XXXX'
          IER=-1
          RETURN
      ENDIF
C=====IF TRACKING ONLY, READ IN MAP FROM FORTRAN UNIT 1 AND
C=====SKIP THE MAPPING SECTION
      IF (ONLYTR) THEN
          CALL RDMAPZP(SSCMP,5,NO,1)
          GO TO 200
      ENDIF
C=====MAPPING SECTION (ZPLIB TRANSLATION FROM OPSTRK & PTRACK) FOLLOWS;
C=====POLYNOMIALLY TRANSLATED SSCTRK VARIABLES HAVE LETTER F PREFIXED
C=====INITIALIZE X,TH,Y,PH MAP INPUT VARIABLES
          CALL ZPOK1(FX,1.0D+0,1,NM)
          CALL ZPOK1(FTH,1.0D+0,2,NM)
          CALL ZPOK1(FY,1.0D+0,3,NM)
          CALL ZPOK1(FPH,1.0D+0,4,NM)
C=====INITIALIZE ZMZI (Z MINUS Z INITIAL) MAP OUTPUT VARIABLE TO ZERO
          CALL ZPZRO(FZMZI,NM)
C=====INITIALIZE MAP INPUT VARIABLES RELATING TO DELP:
C=====K1DELP(V)=K1*DELP(V)
          CALL ZPOK1(FK1DLP,K1,5,NM)
C=====K2DELP(V)=K2*DELP(V)
          CALL ZPOK1(FK2DLP,K2,5,NM)
C=====K3DELP(V)=K3*DELP(V)
          CALL ZPOK1(FK3DLP,K3,5,NM)
C=====RIGITY(V)=1.D+0/(1.D+0+DELP(V))
          CALL ZPOK1(FRIG,1.0D+0,5,NM)
          CALL ZPCADD(1.0D+0,FRIG,FRIG,NM)
          CALL ZPINV(FRIG,NO,FRIG,NO)
C=====INITIALIZATIONS ARE DONE.  CREATE MAP BY TRANSLATING PTRACK.
          DO 100 KK=1,POLELS
C=====REPOL(V)=0.D+0
          CALL ZPZRO(FREPOL,NM)
C=====IMPOL(V)=0.D+0
          CALL ZPZRO(FIMPOL,NM)
          DO 120 I=PPOLES,1,-1

```

```

C=====RETMP(V)=REPOL(V)+ABRE(I,KK)
      CALL ZPCADD(ABRE(I,KK),FREPOL,FRETMP,NM)
C=====IMTMP(V)=IMPOL(V)+ABIM(I,KK)
      CALL ZPCADD(ABIM(I,KK),FIMPOL,FIMTMP,NM)
C=====REPOL(V)=RETMP(V)*X(V)-IMTMP(V)*Y(V)
      CALL ZPMUL(FIMTMP,NO,FY,NO,FTMPO,NO,NOK)
      CALL ZPMUL(FRETMP,NO,FX,NO,FREPOL,NO,NOK)
      CALL ZPSUB(FREPOL,FTMPO,FREPOL,NM)
C=====IMPOL(V)=IMTMP(V)*X(V)+RETMP(V)*Y(V)
      CALL ZPMUL(FRETMP,NO,FY,NO,FTMPO,NO,NOK)
      CALL ZPMUL(FIMTMP,NO,FX,NO,FIMPOL,NO,NOK)
      CALL ZPADD(FIMPOL,FTMPO,FIMPOL,NM)

120      CONTINUE
C=====TH(V)=TH(V)+RIGITY(V)*REPOL(V)
      CALL ZPMUL(FRIG,NO,FREPOL,NO,FTMPO,NO,NOK)
      CALL ZPADD(FTH,FTMPO,FTH,NM)
C=====PH(V)=PH(V)-RIGITY(V)*IMPOL(V)
      CALL ZPMUL(FRIG,NO,FIMPOL,NO,FTMPO,NO,NOK)
      CALL ZPSUB(FPH,FTMPO,FPH,NM)
C=====Z(V)=Z(V)-X(V)*K2-K1*TH(V)-K3DELP(V)
      CALL ZPLIN(FZMZI,-K2,FX,FZMZI,NM)
      CALL ZPLIN(FK3DLP,K1,FTH,FTMPO,NM)
      CALL ZPSUB(FZMZI,FTMPO,FZMZI,NM)
C=====X(V)=X(V)+L0*TH(V)+K1DELP(V)
      CALL ZPLIN(FX,L0,FTH,FX,NM)
      CALL ZPADD(FX,FK1DLP,FX,NM)
C=====Y(V)=Y(V)+L0*PH(V)
      CALL ZPLIN(FY,L0,FPH,FY,NM)
C=====TH(V)=TH(V)+K2DELP(V)
      CALL ZPADD(FTH,FK2DLP,FTH,NM)
      IF (MOD(KK,100).EQ.1) PRINT*,'ELEMENTS THRU ',KK,' MAPPED.'

100      CONTINUE
C=====MAPPING IS COMPLETED -- WRITE OUT MAP TO FORTRAN UNIT 1.
      CALL WRMAPZP(SSCMP,1,5,NO,1)
C      CALL SETXY(NLIST,LIST,0)
C =====TRACKING
200      APERR0=.FALSE.
C =====MAP TRACKING
C      PRINT *, 'IN DELP(1),Z(1)', DELP(1),Z(1)
      DO 20 J=1,PTURNS
        DO 30 V=1,PCASES
          XINPT(1)=X(V)
          XINPT(2)=TH(V)
          XINPT(3)=Y(V)
          XINPT(4)=PH(V)
          XINPT(5)=DELP(V)

```

```

C=====CALL MAP TRACKING ROUTINE (ONE TURN)
      CALL ZPMTRK(SSCMP,1,5,NO,XINPT,YOUTP)
      X(V)=YOUTP(1)
      TH(V)=YOUTP(2)
      Y(V)=YOUTP(3)
      PH(V)=YOUTP(4)
      Z(V)=Z(V)+YOUTP(5)
30      CONTINUE
C IF PARTICLES LEAVE APERTURE THE SUBROUTINE RETURNS
      IER=0
      NLIST=0
      CALL PCHEK(NLIST,LIST,J)
      IF (APERR0) THEN
        APERR0=.FALSE.
        IER=1
        PRINT *, 'OUTSIDE APERTURE'
        RETURN
      ENDIF
C DISPERSION FREE COORDINATES AT RF CAVITY
      DO 50 V=1,PCASES
C DISPERSION FREE COORDINATES AFTER HALF QUAD THEN DISPERSION CORR
C IF TRACK IS REVERSED THEN SIGN OF TWISS MUST CHANGE BEFORE AND
C AFTER PRFACC IN XOUT STATEMENTS
        THOUT(V)=TH(V)+QD/2.*X(V)
        PHOUT(V)=PH(V)-QD/2*Y(V)
        XOUT(V)=X(V)-TWISS(3,1,1)*DELP(V)
        YOUT(V)=Y(V)
C        WRITE (2,*) J,YOUT(V),XOUT(V)
50      CONTINUE
C CALL PRFACC ADDS RF CAVITY ACCELERATION AND ADIABATIC DAMPING TERM
      CALL PRFACC
      DO 60 V=1,PCASES
C DISPERSED COORDINATES AFTER DISPERSION AND -1/2 QUAD
        X(V)=XOUT(V)+TWISS(3,1,1)*DELP(V)
        Y(V)=YOUT(V)
        TH(V)=THOUT(V)-QD/2.*X(V)
        PH(V)=PHOUT(V)+QD/2*Y(V)
60      CONTINUE
20      CONTINUE
      PRINT *, 'OPSMAP:  X1,TH1,P1,Z1', X(1),TH(1),DELP(1),Z(1)
      RETURN
      END

```

```

subroutine zpprep(nv,no,np,nm)
implicit double precision(a-h,o-z)

```

```

parameter (nvm=5,nom=9,nmm=2002,npm=1,
+         nol=nom+1,
+         nov=(nom+2)*nol*nvm,
+         njv=(nvm+1)*nmm,
+         nvp=max(nvm*npm,nmm),
+         nmp=max(nmm*nvm+6,npm),
+         nmw=max(nmm*nvm+6,nol*nvm*npm),
+         nikpm=nol*(nmm-1),
+         navgm=nom/nvm,nrm=nom-navgm*nvm,
+         nkpmx=(navgm+2)**nrm*(navgm+1)**(nvm-nrm),
c +         nkpm=nmm*nkpmx,
+         nkpm=92378,
+         njdm=nmm*nom/(nvm+nom),
+         nvmsq=nvm*nvm)
common /strc1/ nmo(nol)
common /strc2/ nmob(nol)
common /strc3/ nmov(nov)
common /strc4/ jv(njv)
common /strc5/ js(nvm)
common /zptps/ jtpa(nmm)
common /mulp1/ ikp(nikpm)
common /mulb1/ ikb(nikpm)
common /mulp2/ kp(nkpm)
common /mulp3/ lp(nkpm)
common /drvpl/ jd(njdm)
common /mulwk/ wkmul(nvp)
common /divwk/ wkdiv(nmp)
common /conwk/ work(nmw)
common /mtrx1/ aa(nvmsq)
common /mtrx2/ bb(nvmsq)
common /ccsqr/ csqrt(nom)
common /ccinvs/ cinv(nom)
common /ccclns/ cln(nom)
common /ccexps/ cexp(nom)
common /csccoe/ csc(nom)
call zprrp(nv,no,nmm,nol,nov,njv,nikpm,nmw,nkpm,njdm,nm)
return
end

```

```

subroutine zptrkp(nv,no,np,nm)
implicit double precision(a-h,o-z)
parameter (nvm=5,nom=9,nmm=2002,npm=1,
+         nol=nom+1,
+         nov=(nom+2)*nol*nvm,
+         njv=(nvm+1)*nmm,

```



```

+      nvp=max(nvm*npm, nmm),
+      nmp=max(nmm*nvm+6, npm),
+      nmw=max(nmm*nvm+6, nol*nvm*npm))
common /strc1/ nmo(nol)
common /strc2/ nmob(nol)
common /strc3/ nmov(nov)
common /strc4/ jv(njv)
common /strc5/ js(nvm)
common /mulwk/ wkmul(nvp)
common /divwk/ wkdiv(nmp)
common /conwk/ work(nmw)
call trkprp(nv,no,np,nmm,npm,nol,nov,njv,nvp,nmp,nmw,nm)
return
end

```